COSETS OF SUBGROUPS IN MAPLE

Link to: physicspages home page.

To leave a comment or report an error, please use the auxiliary blog and include the title or URL of this post in your comment.

Post date: 5 January 2026.

Reference: *A Gentle Introduction to Group Theory*, Bana Al Subaiei & Muneerah Al Nuwairan, Section 7.4.

Maple has a number of tools for dealing with cosets of subgroups. Here, we'll look at using some of these tools to analyze cosets of permutation groups. We begin with a simple example of the symmetric (permutation) group of 3 numbers $\mathfrak{S}_3$. We've looked at generating this group and its subgroups in an earlier post. The results are

```
with(GroupTheory):
S3 := Symm(3);
Elements(S3);
                 {(1,3,2), (1,3), (2,3), (), (1,2,3), (1,2)}
S3Sub := SubgroupLattice(S3);
S3SubList := convert(S3Sub, 'list');
      S3SubList := [<>, <(2,3)>, <(1,2)>, <(1,3)>, <(1,3,2)>,
                      <(1,3,2),(1,3),(2,3),(1,2)>]
```

The list gives each subgroup in the form of a group generated by a set (hence the $< \ldots >$ notation). To get the actual elements of one of these subgroups, we can use `Elements` as above. For example:

```
S3SubList[2];
                              <(2,3)>
Elements(S3SubList[2]);
                            {(2,3), ()}
```

The set is $\langle(2,3)\rangle$ which generates the group $\{(2,3),e\}$ where $e=()$ is the identity, since $(2,3)$ is its own inverse.

To generate cosets, we can use Maple's `LeftCosets(H,G)` and `RightCosets(H,G)` commands. In these commands, H is a subgroup of G. Note that the subgroup H always appears as the first argument in both commands.

When using these with permutation groups, however, it's important to note that Maple evaluates combinations of permutations from left to right, rather than right to left as is done in most textbooks. The result is that if we want to generate the left cosets referred to in, for example, the reference at the top of this post, we need use the `RightCosets` command, and vice versa. For example, to find the left cosets of the subgroup $H = \{(2,3), e\}$ we use `RightCosets` to get

```
S3_2 := RightCosets(S3SubList[2], S3);
        S3_2 := {<(2,3)>.((1,3)), <(2,3)>.(), <(2,3)>.((1,2))}
```

Using the `Elements` command, we can see the elements of each of these cosets (I've omitted the commas between numbers to save clutter):

```
Elements(S3_2[1]);
                        {(13), (132)}
Elements(S3_2[2]);
                         {(23), ()}
Elements(S3_2[3]);
                        {(123), (12)}
```

Comparing these results with Example 7.4.3 in the above reference, we see that these are the *left* cosets of $H = \{(2,3), e\}$, although we used the `RightCosets(H,G)` command in Maple to generate them.

Similarly, we can use Maple's `LeftCosets(H,G)` command to generate the right cosets:

```
S3_2R := LeftCosets(S3SubList[2], S3);
          S3_2R := {((1,3)).<(2,3)>, ().<(2,3)>, ((1,2)).<(2,3)>}
Elements(S3_2R[1]);
                        {(13), (123)}
Elements(S3_2R[2]);
                         {(23), ()}
Elements(S3_2R[3]);
                        {(132), (12)}
```

These match the right cosets given in Example 7.4.3.

Maple also has the commands `LeftCoset(g,H)` and `RightCoset(H,g)` which find the cosets obtained from a single group element g with respect to a subgroup `H`. Unlike the `LeftCosets(H,G)` and `RightCosets(H,G)` commands where the subgroup H always appears as the first argument, the element g appears as the first (leftmost) argument in `LeftCoset(g,H)` and the second (rightmost) element in `RightCoset(H,g)`.

```
L1 := LeftCoset(Perm([[1, 2, 3]]), S3SubList[2]);
                    L1 := ((1,2,3)) . <(2,3)>
R1 := RightCoset(S3SubList[2], Perm([[1, 2, 3]]));
                    R1 := <(2,3)> . ((1,2,3))
```

```
Elements(L1);
                                {(13), (123)}
Elements(R1);
                                {(123), (12)}
```

In `L1` and `R1`, we use the command `Perm([[1, 2, 3]])` to create the permutation $(1,2,3)$ which is an element of the group `S3` and then calculate its cosets with the subgroup $H=\{(2,3),e\}$. Again, the left and right cosets are swapped, relative to their appearance in the reference book.

PINGBACKS

Pingback: Cosets of subgroups